
Overview

This document lists TTCN changes needed to correct a problem in the ‘iwd-EUTRA-B2009-12_D11wk04’.

Corrections 
Change 1 / Consistent and simplified timing acc. to the prose requirements 
	Function name
	f_TC_7_1_4_11_EUTRA ( )

	Reason for change
	TTCN implementation is not compliant to the prose in which steps 23 and 25 are defined as consecutive retransmissions. But TTCN configures for steps 24-25 a different timing with ‘TransRetransmissionList3’ and ‘Timing2’. Also step 22 is not reflected in the retransmission lists. TC can be simplified in a consistent way.

	Summary of change
	‘TransRetransmissionList3’ is deleted and ‘TransRetransmissionList1’ is extended by 2 elements for steps 22 and 24. Timing for step 25 is also corrected.

	TTCN module
	MAC_714.ttcn 

	MCC160 Comment
	Prose in step 24, does not mandate for UL grant transmission in the next TTI for HARQ process; It is sufficient if it is scheduled in a TTI belonging to HARQ process in future; Hence the current TTCN implementation is valid and conformant to prose. 
The proposed implementation change to schedule step 24 in next TTI for harq process can also be seen as an alternative, but not necessary. Hence proposed changes will not be implemented in TTCN.


Before change:

	  function f_TC_7_1_4_11_EUTRA() runs on EUTRA_PTC
  {
    /* L2/MAC
     * 7.1.4.11 : Correct HARQ process handling
     */
    var integer j;
    var octetstring v_EncodedPDCPPDU, v_EncodedRLCPDU;
    var MAC_PDU_Type v_MAC_PDU;
    var SubFrameTiming_Type v_Timing, v_Timing2;
    var DRB_COMMON_IND v_DRB_COMMON_IND;
    var EUTRA_FDD_TDD_Mode_Type  v_EUTRA_FDD_TDD_Mode ;
    var integer v_RTT := 8; // Round trip time 8 for FDD and 10 for TDD
    var UL_TransRetransmissionList_Type v_UL_TransRetransmissionList1 := { { NewTransmission := true },
                                                                           { ReTransmissionNonAdaptive := true }, // RV 2
                                                                           { ReTransmissionNonAdaptive := true},  // RV 3
                                                                           { ReTransmissionNonAdaptive := true }, // RV 1
                                                                           { ReTransmissionNonAdaptive := true }, // RV 0
                                                                           { ReTransmissionNonAdaptive := true }, // RV 2
                                                                           { ReTransmissionAdaptive := 1 },       // RV 1
                                                                           { ReTransmissionAdaptive := 3 }        // RV 3
    };
    var UL_TransRetransmissionList_Type v_UL_TransRetransmissionList3 := { { ReTransmissionAdaptive := 0 } }; //sic@ R5s100522 sic@
    var template (omit) DRB_ToAddModList v_DRB_ToAddModList := omit;
    timer t_Watchdog := 5.0;
    ///////////////////////////////////////////////////////////////////////////////////
    //      PREAMBLE
    //////////////////////////////////////////////////////////////////////////////////
    f_EUTRA_Init (c1);
    v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD (eutra_Cell1);
    //initialise RTT for TDD
    if (v_EUTRA_FDD_TDD_Mode == TDD)
      {
        v_RTT := 10;
      }
    f_EUTRA_CellConfig_Def (eutra_Cell1);
    f_EUTRA_Preamble(eutra_Cell1, STATE2A_TESTLOOP_ModeA);
    // creates cell and performs registration
    // Disable AS ciphering on UE side.
    f_EUTRA_AS_CipheringAlgorithm_Set(eea0); // set null AS ciphering algorithm
    v_DRB_ToAddModList[0] := cs_508_DRB_ToAddMod_Common(tsc_DRB1,
                                                        cs_508_PDCP_Config_DRB_AM,
                                                        cs_508_RLC_Config_DRB_AM(cds_TX_AM_RLC_t_PollRetransmit(ms250)),
                                                        cs_508_LogicalChannelConfig_DRB_AM);
    f_EUTRA_GenericRbEst_Common(eutra_Cell1, 0, 0, v_DRB_ToAddModList, cs_508_MAC_MainConfig_RBC_maxHARQ_Tx8_Explicit_DrxL, omit, omit, omit); //R5-095104
    f_EUTRA_CloseUE_TestLoopModeA(eutra_Cell1, cs_UE_TestLoopModeA_LB_Setup_NoScaling);
    f_SS_Config1TM_DRB_NoMACHeader(eutra_Cell1, tsc_RbId_DRB1, cs_MAC_TestMode_NoHeaderManipulationDL_UL);
    ///////////////////////////////////////////////////////////////////////////////////
    //      Test Body
    //////////////////////////////////////////////////////////////////////////////////
    f_EUTRA_TestBody_Set(true, layer2); // @sic R5-101050.
    f_EUTRA_StopULGrantTransmission_PUCCH_Sync (eutra_Cell1, cs_TimingInfo_Now); //@sic R5s100522 sic@
    // preconfig SS for PHICH mode
    f_SS_ConfigPhich_TestMode (eutra_Cell1,
                               cs_Phich_ExplicitMode(cs_HARQ_ModeList_4Nack_Ack_2Nack));
    v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1, 200); //@sic R5s100522 sic@
    //@siclog "Step 1" siclog@
    // 1 MAC PDU = 1 RLC PDU;  1 RLC PDU =1 PDCP PDU
    // MAC Header = 1B
    // 1 MAC SDU of 14 => RLC SDU= 12, PDCP SDU = 10
    // Total MAC PDU = 1 + 14  = 15B = 120b (best fit)
    v_EncodedPDCPPDU := f_PDCP_PDU_Encvalue (cs_PDCP_PDU_LongSN(cs_PDCP_DataPdu_LongSN(tsc_B12_SQN0, crs_PDCP_SDU_10B)));
    v_EncodedRLCPDU  := f_RLC_PDU_Encvalue (cs_RLC_AMD_1PDU(tsc_NoPoll, int2bit(0, 10), v_EncodedPDCPPDU));
    DRB.send (cas_DRB_COMMOM_REQ_MAC_PDU (eutra_Cell1,
                                          tsc_RbId_DRB1,
                                          cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),
                                          cs_MAC_PDUList_1 (cs_MAC_PDU_NoPadding_RLCPDU_LCID(tsc_LchId_DRB1, v_EncodedRLCPDU))));
    //@siclog "Step 2 void" siclog@
    //@siclog "Step 3" siclog@
    // Pre config SS for UL Grant of 120 bits
    //Note the function is not used here to guarantee sub frame 9
    //schedule 59 sub frames in future;
    //Subframe 9 means for both FDD and TDD, UE send ul data in sub frame 3.
    v_Timing.SFN.Number := (v_Timing.SFN.Number + 5) mod 1024;
    v_Timing.Subframe.Number := 9;
    f_EUTRA_OneULGrantTransmission( eutra_Cell1,
                                    cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number), //@sic R5s100522 sic@
                                    4, /* Nprb = 2, */
                                    2, /* Itbs = 4
                                          Table 7.1.7.2.1-1 of 36.213: Imcs = 4 -> Itbs = 4   MCS index of table 8.6.1-1 of 36.213
                                          @sic R5s100522: Additional MCC 160 changes sic@ */
                                    v_UL_TransRetransmissionList1 );
    //Preconfig for step 24
    //@siclog "Step 24" siclog@
    //sic@ R5s100522 sic@
    v_Timing2 := f_EUTRA_TimingInfoAdd (v_Timing, 10*v_RTT);
    f_EUTRA_OneULGrantTransmission(eutra_Cell1,
                                   cs_TimingInfo(v_Timing2.SFN.Number, v_Timing2.Subframe.Number),
                                   4, 2,//Nprb = 2, Itbs = 4
                                   v_UL_TransRetransmissionList3);
    //@siclog "Steps 17 - 20" siclog@
    // Preconfig SS
    v_Timing := f_GetSubFrameNumForULData(v_EUTRA_FDD_TDD_Mode, v_Timing);
    //@siclog "Step 4" siclog@
    t_Watchdog.start;
    alt {
       []DRB.receive (car_DRB_COMMON_IND_MAC (eutra_Cell1,
                                              tsc_RbId_DRB1,
                                              cr_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),
                                              {cr_MAC_PDU_NoPadding_RLCPDU_LCID(tsc_LchId_DRB1, cr_Octet14_Any)})) -> value v_DRB_COMMON_IND
         {
           f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Step 4");
           //v_Timing := v_DRB_COMMON_IND.Common.TimingInfo.SubFrame;
           //v_EncodedMACPDU:= v_DRB_COMMON_IND.U_Plane.SubframeData.PduSduList.MacPdu[0].SduList[0];
           v_MAC_PDU:= v_DRB_COMMON_IND.U_Plane.SubframeData.PduSduList.MacPdu[0];
           v_Timing := f_EvaluateNext_UL_TransmissionForHARQ_Process(v_Timing, v_EUTRA_FDD_TDD_Mode); // Next TTI.
        }
    };
    for (j:=0; j<4; j:= j+1)
      {
        //@siclog "Steps 6 - 8 - 10 - 12" siclog@
        //@siclog "Step 14 void" siclog@
        alt {
          []DRB.receive (car_DRB_COMMON_IND_MAC (eutra_Cell1,
                                                 tsc_RbId_DRB1,
                                                 cr_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),
                                                 {v_MAC_PDU})) -> value v_DRB_COMMON_IND
            {
              f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Steps 6 - 8 - 10 - 12");
              v_Timing := f_EvaluateNext_UL_TransmissionForHARQ_Process(v_Timing, v_EUTRA_FDD_TDD_Mode); // Next TTI.
            }
        };
        //@siclog "Steps 7 - 9 - 11" siclog@
        //@siclog "Step 13 void" siclog@
        //@siclog "Step 15" siclog@
        // Automatically taken care off
      }// End of For loop
    // Note v_Timing holds timing for step 16;
    //Timing for step 18
    v_Timing := f_EvaluateNext_UL_TransmissionForHARQ_Process(v_Timing, v_EUTRA_FDD_TDD_Mode);
    //@siclog "Steps 16 - 18" siclog@
    alt {
      []DRB.receive (car_DRB_COMMON_IND_MAC (eutra_Cell1,
                                             tsc_RbId_DRB1,
                                             cr_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),
                                             {v_MAC_PDU}))
        {
          f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "step 18");
          v_Timing := f_EvaluateNext_UL_TransmissionForHARQ_Process(v_Timing, v_EUTRA_FDD_TDD_Mode);
        }
      []DRB.receive (car_DRB_COMMON_IND_MAC (eutra_Cell1,
                                             tsc_RbId_DRB1,
                                             cr_TimingInfo_Any,
                                             {v_MAC_PDU})) -> value v_DRB_COMMON_IND
        {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 16: UE is Retrasmitting MAC PDU");
          // it is assumed SS will route MAC PDU corresponding to step 16[if any] before step 18
          // if step 18 is executed, then it is confirmed step 16 is passed
        }
    };
    //@siclog "Step 19" siclog@
    // Automatically taken care off
    //@siclog "Step 21" siclog@
    DRB.receive (car_DRB_COMMON_IND_MAC(eutra_Cell1,
                                        tsc_RbId_DRB1,
                                        cr_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),
                                        {v_MAC_PDU}));
    f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Step 21");
    //@siclog "Step 22" siclog@
    // Automatically taken care off
    //@siclog "Step 23, 25" siclog@
    //sic@ R5s100522 sic@
    v_Timing := f_EvaluateNext_UL_TransmissionForHARQ_Process(v_Timing, v_EUTRA_FDD_TDD_Mode);
    v_Timing2 := f_GetSubFrameNumForULData(v_EUTRA_FDD_TDD_Mode, v_Timing2);
    alt {
      []DRB.receive (car_DRB_COMMON_IND_MAC (eutra_Cell1,
                                             tsc_RbId_DRB1,
                                             cr_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),
                                             {v_MAC_PDU}))
        {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 23: UE is Retransmitting MAC PDU");
          t_Watchdog.stop;
        }
      [] DRB.receive (car_DRB_COMMON_IND_MAC (eutra_Cell1,
                                              tsc_RbId_DRB1,
                                              cr_TimingInfo(v_Timing2.SFN.Number, v_Timing2.Subframe.Number),
                                              {v_MAC_PDU}))
        {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 25: UE is Retransmitting MAC PDU");
          t_Watchdog.stop;
        }
      [] DRB.receive (car_DRB_COMMON_IND_MAC (eutra_Cell1,
                                              tsc_RbId_DRB1,
                                              cr_TimingInfo(v_Timing2.SFN.Number, v_Timing2.Subframe.Number),
                                              {cr_MAC_PDU_NoSDU}))
        {
          t_Watchdog.stop;
        }
    };
    //@siclog "Step 25a" siclog@
    //@sic R5-104162 sic@
    f_RLC_Status_PDU_Tx (tsc_LchId_DRB1, 1);
    f_SS_ConfigPhich_TestMode (eutra_Cell1, cs_PhichTestModeNormal);
    //@siclog "Step 26" siclog@
    //sic@ R5s100522 sic@


After change:

	  function f_TC_7_1_4_11_EUTRA() runs on EUTRA_PTC
  {
    /* L2/MAC
     * 7.1.4.11 : Correct HARQ process handling
     */
    var integer j;
    var octetstring v_EncodedPDCPPDU, v_EncodedRLCPDU;
    var MAC_PDU_Type v_MAC_PDU;
    var SubFrameTiming_Type v_Timing, v_Timing2;
    var DRB_COMMON_IND v_DRB_COMMON_IND;
    var EUTRA_FDD_TDD_Mode_Type  v_EUTRA_FDD_TDD_Mode ;
    var integer v_RTT := 8; // Round trip time 8 for FDD and 10 for TDD
    var UL_TransRetransmissionList_Type v_UL_TransRetransmissionList1 := { { NewTransmission := true },
                                                                           { ReTransmissionNonAdaptive := true }, // RV 2
                                                                           { ReTransmissionNonAdaptive := true},  // RV 3
                                                                           { ReTransmissionNonAdaptive := true }, // RV 1
                                                                           { ReTransmissionNonAdaptive := true }, // RV 0
                                                                           { ReTransmissionNonAdaptive := true }, // RV 2
                                                                           { ReTransmissionAdaptive := 1 },       // RV 1
                                                                           { ReTransmissionAdaptive := 3 },       // RV 3
                                                                           { ReTransmissionNonAdaptive := true }, // RV ?
                                                                           { ReTransmissionAdaptive := 0 }        // RV 0
    };
    // var UL_TransRetransmissionList_Type v_UL_TransRetransmissionList3 := { { ReTransmissionAdaptive := 0 } }; //sic@ R5s100522 sic@
    var template (omit) DRB_ToAddModList v_DRB_ToAddModList := omit;
    timer t_Watchdog := 5.0;
    ///////////////////////////////////////////////////////////////////////////////////
    //      PREAMBLE
    //////////////////////////////////////////////////////////////////////////////////
    f_EUTRA_Init (c1);
    v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD (eutra_Cell1);
    //initialise RTT for TDD
    if (v_EUTRA_FDD_TDD_Mode == TDD)
      {
        v_RTT := 10;
      }
    f_EUTRA_CellConfig_Def (eutra_Cell1);
    f_EUTRA_Preamble(eutra_Cell1, STATE2A_TESTLOOP_ModeA);
    // creates cell and performs registration
    // Disable AS ciphering on UE side.
    f_EUTRA_AS_CipheringAlgorithm_Set(eea0); // set null AS ciphering algorithm
    v_DRB_ToAddModList[0] := cs_508_DRB_ToAddMod_Common(tsc_DRB1,
                                                        cs_508_PDCP_Config_DRB_AM,
                                                        cs_508_RLC_Config_DRB_AM(cds_TX_AM_RLC_t_PollRetransmit(ms250)),
                                                        cs_508_LogicalChannelConfig_DRB_AM);
    f_EUTRA_GenericRbEst_Common(eutra_Cell1, 0, 0, v_DRB_ToAddModList, cs_508_MAC_MainConfig_RBC_maxHARQ_Tx8_Explicit_DrxL, omit, omit, omit); //R5-095104
    f_EUTRA_CloseUE_TestLoopModeA(eutra_Cell1, cs_UE_TestLoopModeA_LB_Setup_NoScaling);
    f_SS_Config1TM_DRB_NoMACHeader(eutra_Cell1, tsc_RbId_DRB1, cs_MAC_TestMode_NoHeaderManipulationDL_UL);
    ///////////////////////////////////////////////////////////////////////////////////
    //      Test Body
    //////////////////////////////////////////////////////////////////////////////////
    f_EUTRA_TestBody_Set(true, layer2); // @sic R5-101050.
    f_EUTRA_StopULGrantTransmission_PUCCH_Sync (eutra_Cell1, cs_TimingInfo_Now); //@sic R5s100522 sic@
    // preconfig SS for PHICH mode
    f_SS_ConfigPhich_TestMode (eutra_Cell1,
                               cs_Phich_ExplicitMode(cs_HARQ_ModeList_4Nack_Ack_2Nack));
    v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1, 200); //@sic R5s100522 sic@
    //@siclog "Step 1" siclog@
    // 1 MAC PDU = 1 RLC PDU;  1 RLC PDU =1 PDCP PDU
    // MAC Header = 1B
    // 1 MAC SDU of 14 => RLC SDU= 12, PDCP SDU = 10
    // Total MAC PDU = 1 + 14  = 15B = 120b (best fit)
    v_EncodedPDCPPDU := f_PDCP_PDU_Encvalue (cs_PDCP_PDU_LongSN(cs_PDCP_DataPdu_LongSN(tsc_B12_SQN0, crs_PDCP_SDU_10B)));
    v_EncodedRLCPDU  := f_RLC_PDU_Encvalue (cs_RLC_AMD_1PDU(tsc_NoPoll, int2bit(0, 10), v_EncodedPDCPPDU));
    DRB.send (cas_DRB_COMMOM_REQ_MAC_PDU (eutra_Cell1,
                                          tsc_RbId_DRB1,
                                          cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),
                                          cs_MAC_PDUList_1 (cs_MAC_PDU_NoPadding_RLCPDU_LCID(tsc_LchId_DRB1, v_EncodedRLCPDU))));
    //@siclog "Step 2 void" siclog@
    //@siclog "Step 3" siclog@
    // Pre config SS for UL Grant of 120 bits
    //Note the function is not used here to guarantee sub frame 9
    //schedule 59 sub frames in future;
    //Subframe 9 means for both FDD and TDD, UE send ul data in sub frame 3.
    v_Timing.SFN.Number := (v_Timing.SFN.Number + 5) mod 1024;
    v_Timing.Subframe.Number := 9;
    f_EUTRA_OneULGrantTransmission( eutra_Cell1,
                                    cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number), //@sic R5s100522 sic@
                                    4, /* Nprb = 2, */
                                    2, /* Itbs = 4
                                          Table 7.1.7.2.1-1 of 36.213: Imcs = 4 -> Itbs = 4   MCS index of table 8.6.1-1 of 36.213
                                          @sic R5s100522: Additional MCC 160 changes sic@ */
                                    v_UL_TransRetransmissionList1 );
    //Preconfig for step 24
    //@siclog "Step 24" siclog@
    //sic@ R5s100522 sic@
    /*
    v_Timing2 := f_EUTRA_TimingInfoAdd (v_Timing, 10*v_RTT);

    f_EUTRA_OneULGrantTransmission(eutra_Cell1,

                                   cs_TimingInfo(v_Timing2.SFN.Number, v_Timing2.Subframe.Number),

                                   4, 2,//Nprb = 2, Itbs = 4

                                   v_UL_TransRetransmissionList3);
    */
    //@siclog "Steps 17 - 20" siclog@
    // Preconfig SS
    v_Timing := f_GetSubFrameNumForULData(v_EUTRA_FDD_TDD_Mode, v_Timing);
    //@siclog "Step 4" siclog@
    t_Watchdog.start;
    alt {
       []DRB.receive (car_DRB_COMMON_IND_MAC (eutra_Cell1,
                                              tsc_RbId_DRB1,
                                              cr_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),
                                              {cr_MAC_PDU_NoPadding_RLCPDU_LCID(tsc_LchId_DRB1, cr_Octet14_Any)})) -> value v_DRB_COMMON_IND
         {
           f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Step 4");
           //v_Timing := v_DRB_COMMON_IND.Common.TimingInfo.SubFrame;
           //v_EncodedMACPDU:= v_DRB_COMMON_IND.U_Plane.SubframeData.PduSduList.MacPdu[0].SduList[0];
           v_MAC_PDU:= v_DRB_COMMON_IND.U_Plane.SubframeData.PduSduList.MacPdu[0];
           v_Timing := f_EvaluateNext_UL_TransmissionForHARQ_Process(v_Timing, v_EUTRA_FDD_TDD_Mode); // Next TTI.
        }
    };
    for (j:=0; j<4; j:= j+1)
      {
        //@siclog "Steps 6 - 8 - 10 - 12" siclog@
        //@siclog "Step 14 void" siclog@
        alt {
          []DRB.receive (car_DRB_COMMON_IND_MAC (eutra_Cell1,
                                                 tsc_RbId_DRB1,
                                                 cr_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),
                                                 {v_MAC_PDU})) -> value v_DRB_COMMON_IND
            {
              f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Steps 6 - 8 - 10 - 12");
              v_Timing := f_EvaluateNext_UL_TransmissionForHARQ_Process(v_Timing, v_EUTRA_FDD_TDD_Mode); // Next TTI.
            }
        };
        //@siclog "Steps 7 - 9 - 11" siclog@
        //@siclog "Step 13 void" siclog@
        //@siclog "Step 15" siclog@
        // Automatically taken care off
      }// End of For loop
    // Note v_Timing holds timing for step 16;
    //Timing for step 18
    v_Timing := f_EvaluateNext_UL_TransmissionForHARQ_Process(v_Timing, v_EUTRA_FDD_TDD_Mode);
    //@siclog "Steps 16 - 18" siclog@
    alt {
      []DRB.receive (car_DRB_COMMON_IND_MAC (eutra_Cell1,
                                             tsc_RbId_DRB1,
                                             cr_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),
                                             {v_MAC_PDU}))
        {
          f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "step 18");
          v_Timing := f_EvaluateNext_UL_TransmissionForHARQ_Process(v_Timing, v_EUTRA_FDD_TDD_Mode);
        }
      []DRB.receive (car_DRB_COMMON_IND_MAC (eutra_Cell1,
                                             tsc_RbId_DRB1,
                                             cr_TimingInfo_Any,
                                             {v_MAC_PDU})) -> value v_DRB_COMMON_IND
        {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 16: UE is Retrasmitting MAC PDU");
          // it is assumed SS will route MAC PDU corresponding to step 16[if any] before step 18
          // if step 18 is executed, then it is confirmed step 16 is passed
        }
    };
    //@siclog "Step 19" siclog@
    // Automatically taken care off
    //@siclog "Step 21" siclog@
    DRB.receive (car_DRB_COMMON_IND_MAC(eutra_Cell1,
                                        tsc_RbId_DRB1,
                                        cr_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),
                                        {v_MAC_PDU}));
    f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Step 21");
    //@siclog "Step 22" siclog@
    // Automatically taken care off
    //@siclog "Step 23, 25" siclog@
    //sic@ R5s100522 sic@
    v_Timing := f_EvaluateNext_UL_TransmissionForHARQ_Process(v_Timing, v_EUTRA_FDD_TDD_Mode);
    //v_Timing2 := f_GetSubFrameNumForULData(v_EUTRA_FDD_TDD_Mode, v_Timing2);
    v_Timing2 := f_EvaluateNext_UL_TransmissionForHARQ_Process(v_Timing, v_EUTRA_FDD_TDD_Mode);
    alt {
      []DRB.receive (car_DRB_COMMON_IND_MAC (eutra_Cell1,
                                             tsc_RbId_DRB1,
                                             cr_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),
                                             {v_MAC_PDU}))
        {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 23: UE is Retransmitting MAC PDU");
          t_Watchdog.stop;
        }
      [] DRB.receive (car_DRB_COMMON_IND_MAC (eutra_Cell1,
                                              tsc_RbId_DRB1,
                                              cr_TimingInfo(v_Timing2.SFN.Number, v_Timing2.Subframe.Number),
                                              {v_MAC_PDU}))
        {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 25: UE is Retransmitting MAC PDU");
          t_Watchdog.stop;
        }
      [] DRB.receive (car_DRB_COMMON_IND_MAC (eutra_Cell1,
                                              tsc_RbId_DRB1,
                                              cr_TimingInfo(v_Timing2.SFN.Number, v_Timing2.Subframe.Number),
                                              {cr_MAC_PDU_NoSDU}))
        {
          t_Watchdog.stop;
        }
    };
    //@siclog "Step 25a" siclog@
    //@sic R5-104162 sic@
    f_RLC_Status_PDU_Tx (tsc_LchId_DRB1, 1);
    f_SS_ConfigPhich_TestMode (eutra_Cell1, cs_PhichTestModeNormal);
    //@siclog "Step 26" siclog@
    //sic@ R5s100522 sic@

